Programme ISO 7933

Note : For a certain range of exposure conditions, the model predicts a steady state rectal temperature near the 38°C limit. The thermal stress is obviously the same whether this steady state value is just below 38°C (in which case there is no limitation of the duration of exposure for heat accumulation) or slightly above (in which case the allowable exposure time could be short). In order to allow the user to be aware of these borderline situations and to make professional judgements regarding the duration of exposure, it is strongly recommended that the programmes not only provide the final values of accumulated water loss and rectal temperature, but make it possible to analyze, graphically or otherwise, the evolution of these parameters over time. This becomes essential when the exposure is variable over time.

```
' Predicted Heat Strain (PHS) model
' EXPONENTIAL AVERAGING CONSTANTS
   ConstTeq = Exp(-1 / 10):          ' Core temperature as a function of M: time constant: 10 min
   ConstTsk = Exp(-1 / 3):           ' Skin Temperature: time constant: 3 min
   ConstSW = Exp(-1 / 10):           ' Sweat rate: time constant: 10 min
' INPUT OF THE MEAN CHARACTERISTICS OF THE PERSONS
' The user must make sure at this point in the programme that the following parameters are available.
' Standard values must be replaced by actual values.
   Weight = 75:    ' Body mass  kilogrammes
   Height = 1.8:   ' Body height   metres
   Accl = 1:                 ' =1 if acclimatized person, =0 otherwise
   Drink = 1:                ' Water replacement: =1 if the persons can drink freely, =0 otherwise
' COMPUTATION OF DERIVED PARAMETERS
   Adu = 0.202 * Weight ^ 0.425 * Height ^ 0.725:      ' Body surface area m²
   aux = 3490 * Weight / Adu:                                  ' Heat for 1°C increase of the body per m² of body surface
   SWmax = 400:   If Accl = 1 Then SWmax = 500:   ' Maximum evaporative capacity
   wmax = 0.85:   If Accl = 1 Then wmax = 1:                   ' Maximum wettedness
   Dmax = 0.05 * Weight * 1000:                        ' Maximum water loss in grams
      If Drink = 0 Then Dmax = 0.03 * Weight * 1000:           ' if no free drinking
' INPUT OF THE PRIMARY PARAMETERS
' The user shall make sure that, at this point in the programme, the following parameters are available.
' In order for the user to test rapidly the programme, the data for the first case in Annex F of ISO 7933 (2020) are
introduced.
   Duration = 480:           ' Duration of the work sequence, in minutes
   Ta = 40:                  ' Air temperature, in °C
   Tg = 40:                  ' Black globe temperature, in °C
   Diam = 15:                ' Diameter of the black globe, in cm
   Va = 0.3:                 ' Air velocity, in metres per second
      Tr = ((Tg + 273) ^ 4 + 1.1579 * 10 ^ 8 / 0.95 / (Diam / 100) ^ 0.4 * Va ^ 0.6 * (Tg - Ta)) ^ 0.25 - 273
   RH = 35:                  ' Relative humidity
' Partial water vapour pressure kilopascals
      Pa = 0.6105 * Exp(17.27 * Ta / (Ta + 237.3)) * RH / 100:
   M = 300:                  ' Metabolic rate, in watts
      Met = M / Adu:         ' Metabolic rate, in watts per square metre
   Work = 0:                 ' Effective mechanical power, in watts per square metre
   Icl = 0.5:                ' Static thermal insulation, in clo
   imst = 0.38:              ' Static moisture permeability index
   Fr = 0.42                  ‘ Reflection coefficient for different special materials

' Effective radiating area of the body
   Posture = 1:              ' Posture = 1 standing, =2 sitting, =3 crouching
      If Posture = 1 Then Ardu = 0.77
      If Posture = 2 Then Ardu = 0.7
      If Posture = 3 Then Ardu = 0.67
' Reflective clothing
   Ap = 0.54:                ' Fraction of the body surface covered by the reflective clothing
   Ecl = 0.97:               ' Emissivity of the clothed body surface (by default: Ecl=0.97)
' Displacements
```

```
   defspeed = 0:  ' =1 if walking speed entered, =0 otherwise
   Walksp = 0:                ' Walking speed, m/s
   defdir = 0:                ' =1 if walking direction entered, 0 otherwise
   THETA = 0:                 ' Angle between walking direction and wind direction degrees
' CLOTHING INFLUENCE ON EXCHANGE COEFFICIENTS
   Iclst = Icl * 0.155:       ' Static clothing insulation
   fcl = 1 + 0.3 * Icl:       ' Clothing area factor
   Iast = 0.111:   ' Static boundary layer thermal insulation in quiet air
   Itotst = Iclst + Iast / fcl:  ' Total static insulation
' Relative velocities due to air velocity and movements
   If defspeed > 0 Then
      If defdir = 1 Then
         Var = Abs(Va - Walksp * Cos(3.14159 * THETA / 180)):        ' Unidirectional walking
      Else
         If Va < Walksp Then Var = Walksp Else Var = Va:             'Omni-directional walking
      End If
   Else
      Walksp = 0.0052 * (Met - 58)
         If Walksp > 0.7 Then Walksp = 0.7:  'Stationary or undefined speed
      Var = Va
   End If
' Dynamic clothing insulation
   Vaux = Var:  If Var > 3 Then Vaux = 3
   Waux = Walksp:  If Walksp > 1.5 Then Waux = 1.5
' Clothing insulation correction for wind (Var) and walking (Walksp)
   CORcl = 1.044 * Exp((0.066 * Vaux - 0.398) * Vaux + (0.094 * Waux - 0.378) * Waux)
      If CORcl > 1 Then CORcl = 1
   CORia = Exp((0.047 * Vaux - 0.472) * Vaux + (0.117 * Waux - 0.342) * Waux)
      If CORia > 1 Then CORia = 1
   CORtot = CORcl
      If Icl <= 0.6 Then CORtot = ((0.6 - Icl) * CORia + Icl * CORcl) / 0.6
   Itotdyn = Itotst * CORtot
   Iadyn = CORia * Iast
   Icldyn = Itotdyn - Iadyn / fcl
' Dynamic evaporative resistance
' Correction for wind and walking
   CORe = (2.6 * CORtot - 6.5) * CORtot + 4.9
   imdyn = imst * CORe: If imdyn > 0.9 Then imdyn = 0.9
   Rtdyn = Itotdyn / imdyn / 16.7
' INITIALISATION OF THE VARIABLES OF THE PROGRAMME
   Tre = 36.8:                ' Initial rectal temperature, °C
   Tcr = 36.8:                ' Initial core temperature, °C
   Tsk = 34.1:                ' Initial skin temperature, °C
   Tcreq = 36.8:  ' Initial core temperature associated to M, °C
   TskTcrwg = 0.3:            ' Initial skin – core weighting
   SWp = 0:                   ' Initial sweat rate, W/m²
   SWtot = 0:                 ' Initial total sweat rate, W/m²
   Dlimtcr = 999:             '  Allowable exposure time due to increase in temperature, min
   Dlimloss = 999:            ' Allowable exposure time due to excessive water loss, min
' ITERATION OF THE PROGRAMME
For Time = 1 To Duration
' Initialisation min per min
' value at beginning of time i = final value at time (i-1)
   Tre0 = Tre: Tcr0 = Tcr: Tsk0 = Tsk: Tcreq0 = Tcreq: TskTcrwg0 = TskTcrwg
' Equilibrium core temperature associated to the metabolic rate
   Tcreqm = 0.0036 * Met + 36.6
' Core temperature at this minute, by exponential averaging
```

```
    Tcreq = Tcreq0 * ConstTeq + Tcreqm * (1 - ConstTeq)
' Heat storage associated with this core temperature increase during the last minute
    dStoreq = aux/60 * (Tcreq - Tcreq0) * (1 - TskTcrwg0)
' SKIN TEMPERATURE PREDICTION
' Skin Temperature in equilibrium
' Clothed model
    Tskeqcl = 12.165 + 0.02017 * Ta + 0.04361 * Tr + 0.19354 * Pa - 0.25315 * Va
    Tskeqcl = Tskeqcl + 0.005346 * Met + 0.51274 * Tre
' Nude model
    Tskeqnu = 7.191 + 0.064 * Ta + 0.061 * Tr + 0.198 * Pa - 0.348 * Va
    Tskeqnu = Tskeqnu + 0.616 * Tre
' Value at this minute, as a function of the clothing insulation
        If Icl >= 0.6 Then Tskeq = Tskeqcl: GoTo Tsk
        If Icl <= 0.2 Then Tskeq = Tskeqnu: GoTo Tsk
' Interpolation between the values for clothed and nude person, if 0.2 < clo < 0.6
    Tskeq = Tskeqnu + 2.5 * (Tskeqcl - Tskeqnu) * (Icl - 0.2)
' Skin Temperature at this minute, by exponential averaging
Tsk:
    Tsk = Tsk0 * ConstTsk + Tskeq * (1 - ConstTsk)
        If Time = 1 Then Tsk = Tskeq
' Saturated water vapour pressure at the surface of the skin
    Psk = 0.6105 * Exp(17.27 * Tsk / (Tsk + 237.3))
' Mean temperature of the clothing: Tcl
    Z = 3.5 + 5.2 * Var
        If Var > 1 Then Z = 8.7 * Var ^ 0.6
    auxR = 0.0000000567 * Ardu
    Eclr = (1 - Ap) * 0.97 + Ap * (1-Fr)
    Tcl = Tr + 0.1
Tcl:
' convection coefficient
    Hc = 2.38 * Abs(Tcl - Ta) ^ 0.25
        If Z > Hc Then Hc = Z
' Radiation coefficient
    HR = Eclr * auxR * ((Tcl + 273) ^ 4 - (Tr + 273) ^ 4) / (Tcl - Tr)
    Tcl1 = ((fcl * (Hc * Ta + HR * Tr) + Tsk / Icldyn)) / (fcl * (Hc + HR) + 1 / Icldyn)
    If Abs(Tcl - Tcl1) > 0.001 Then Tcl = (Tcl + Tcl1) / 2:     GoTo Tcl
' HEAT EXCHANGES
    Texp = 28.56 + 0.115 * Ta + 0.641 * Pa:   ' Temperature of the expired air
    Cres = 0.001516 * Met * (Texp - Ta):        ' Heat exchanges through respiratory convection
    Eres = 0.00127 * Met * (59.34 + 0.53 * Ta - 11.63 * Pa):      ' through respiratory evaporation
    Conv = fcl * Hc * (Tcl - Ta):                               ' Heat exchange through convection
    Rad = fcl * HR * (Tcl - Tr):                                ' Heat exchange through radiation
    Emax = (Psk - Pa) / Rtdyn:                                  ' Maximum Evaporation Rate
    Ereq = Met - dStoreq - Work - Cres - Eres - Conv - Rad:      ' Required Evaporation Rate
' INTERPRETATION
    wreq = Ereq / Emax:                  ' Required wettedness
' If no evaporation required: no sweat rate
        If Ereq <= 0 Then Ereq = 0: SWreq = 0: GoTo SWp
' If evaporation is not possible, sweat rate is maximum
        If Emax <= 0 Then Emax = 0: SWreq = SWmax: GoTo SWp
' If required wettedness greater than 1.7: sweat rate is maximum
        If wreq >= 1.7 Then wreq = 1.7: SWreq = SWmax: GoTo SWp
    Eveff = 1 - wreq ^ 2 / 2:                         ' Required evaporation efficiency
        If wreq > 1 Then Eveff = (2 - wreq) ^ 2 / 2
    SWreq = Ereq / Eveff:                ' Required Sweat Rate
        If SWreq > SWmax Then SWreq = SWmax:  ' limited to the maximum evaporative capacity
SWp:
```

```
' Predicted Sweat Rate, by exponential averaging
   SWp = SWp * ConstSW + SWreq * (1 - ConstSW)
      If SWp <= 0 Then Ep = 0: SWp = 0: GoTo Storage
' Predicted Evaporation Rate
   k = Emax / SWp
   wp = 1
      If k >= 0.5 Then wp = -k + Sqr(k * k + 2)
      If wp > wmax Then wp = wmax
   Ep = wp * Emax
' Heat Storage
Storage:
   dStorage = Ereq - Ep + dStoreq
' PREDICTION OF THE CORE TEMPERATURE
   Tcr1 = Tcr0
TskTcr:
' Skin - Core weighting
   TskTcrwg = 0.3 - 0.09 * (Tcr1 - 36.8)
      If TskTcrwg > 0.3 Then TskTcrwg = 0.3
      If TskTcrwg < 0.1 Then TskTcrwg = 0.1
   Tcr = dStorage / (aux/60) + Tsk0 * TskTcrwg0 / 2 - Tsk * TskTcrwg / 2
   Tcr = (Tcr + Tcr0 * (1 - TskTcrwg0 / 2)) / (1 - TskTcrwg / 2)
   If Abs(Tcr - Tcr1) > 0.001 Then Tcr1 = (Tcr1 + Tcr) / 2:  GoTo TskTcr
' PREDICTION OF THE RECTAL TEMPERATURE
   Tre = Tre0 + (2 * Tcr - 1.962 * Tre0 - 1.31) / 9
' TOTAL WATER LOSS RATE AFTER THE MINUTE (in W m-²)
   SWtot = SWtot + SWp + Eres:              ' Total evaporation loss in watts per m²
   SWtotg = SWtot * 2.67 * Adu / 1.8 / 60:  ' Total water loss in grams
' COMPUTATION OF THE DURATION LIMIT OF EXPOSURE DLE IN MIN
' DLE for water loss, 95 % of the working population, in min
   If Dlimloss = 999 And SWtotg >= Dmax Then Dlimloss = Time
' DLE for heat storage, in min
   If Dlimtcr = 999 And Tre >= 38 Then Dlimtcr = Time
' End of loop on duration
Next Time
```